In[24]:=
```
Clear[a, b, coelho, eq1, eq2, eq3, f, fig,
 g1, g2, g3, g4, g5, L, retas, solucoes, tabela, x, x0]
```

Um carro anda por uma rodovia que tem o formato da curva $y = x^6 - 7x^4 + 11x^2 + x$. Um coelho se encontra parado no ponto $P = (-1, 2)$, fora da rodovia. É noite e o carro tem seus faróis acesos. Em quais pontos da rodovia o carro iluminará o coelho?

In[25]:=
```
f[x_] = x^6 - 7 x^4 + 11 x^2 + x   (* esta é a função que dá y=f(x) da rodovia *)
```

Out[25]= $x + 11x^2 - 7x^4 + x^6$

In[26]:=
```
g1 = Plot[f[x], {x, -2.5, 2.5}, PlotRange → {-7, 7}
 ]    (* gráfico da rodovia  *)
g2 = ListPlot[{{1, -2}}, PlotStyle → {Black, Large}]
(* ponto em que o coelho está  *)
Show[g1, g2]
```

Out[26]=


Out[27]=


Out[28]=


In[29]:=
```
L[x_] = a x + b    (* função que dá a reta y=L(x) que corresponde
     ao traçado do farol. Note que temos a e b a determinar!  *)
```

Out[29]= b + a x

In[30]:=
```
(*  Temos o problema de determinar a e b da reta acima,
bem como o ponto (x0,f(x0)) em que o carro estará quando o coelho for iluminado;
temos assim 3 incógnitas. Dessa forma, precisamos de três condições (ou equações)
 para determinar tais incógnitas. Essas 3 equações são as seguintes: *)
```

In[31]:=
```
eq1 = L[1] == -2        (*  a reta deve passar pelo coelho em (-2,1)  *)
eq2 = L[x0] == f[x0]    (*  a reta deve passar pelo ponto (x0,f(x0)) do gráfico,
com x0 a determinar  *)
eq3 = a == f'[x0]       (*  a reta deve ter coeficiente angular
   f'(x0) nesse ponto (x0,f(x0)) do gráfico, com x0 a determinar  *)
```

Out[31]= $a + b == -2$

Out[32]= $b + a\,x0 == x0 + 11\,x0^2 - 7\,x0^4 + x0^6$

Out[33]= $a == 1 + 22\,x0 - 28\,x0^3 + 6\,x0^5$

In[34]:=
```
Solve[{eq1, eq2, eq3}, {a, b, x0}]
(* Resolve analiticamente essas 3 equações para essas 3 incógnitas: a, b e x0  *)
```

Out[34]=
```
{{a → 1 + 22 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 1] -
    28 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 1]^3 +
    6 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 1]^5,
  b → -3 - 22 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 1] +
    28 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 1]^3 -
    6 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 1]^5,
  x0 → Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 1]},
 {a → 1 + 22 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 2] -
    28 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 2]^3 +
    6 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 2]^5,
  b → -3 - 22 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 2] +
    28 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 2]^3 -
    6 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 2]^5,
  x0 → Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 2]},
 {a → 1 + 22 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 3] -
    28 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 3]^3 +
    6 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 3]^5,
  b → -3 - 22 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 3] +
    28 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 3]^3 -
    6 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 3]^5,
  x0 → Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 3]},
 {a → 1 + 22 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 4] -
```

```
    28 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 4]^3 +
    6 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 4]^5,
  b → -3 - 22 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 4] +
    28 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 4]^3 -
    6 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 4]^5,
  x0 → Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 4]},
 {a → 1 + 22 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 5] -
    28 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 5]^3 +
    6 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 5]^5,
  b → -3 - 22 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 5] +
    28 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 5]^3 -
    6 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 5]^5,
  x0 → Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 5]},
 {a → 1 + 22 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 6] -
    28 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 6]^3 +
    6 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 6]^5,
  b → -3 - 22 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 6] +
    28 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 6]^3 -
    6 Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 6]^5,
  x0 → Root[-3 - 22 #1 + 11 #1^2 + 28 #1^3 - 21 #1^4 - 6 #1^5 + 5 #1^6 &, 6]}}
```

```
In[35]:= (* O resultado acima é horrível de ser
          interpretado! Vamos proceder então numericamente *)

In[36]:=

In[37]:= NSolve[{eq1, eq2, eq3}, {a, b, x0}]
         (* Resolve numericamente essas 3 equações para essas 3 incógnitas, a, b e x0  *)

Out[37]= {{a → 4.0606 - 25.286 i, b → -6.0606 + 25.286 i, x0 → 1.0959 + 0.531466 i},
          {a → 4.0606 + 25.286 i, b → -6.0606 - 25.286 i, x0 → 1.0959 - 0.531466 i},
          {a → 1.50863, b → -3.50863, x0 → -1.91048},
          {a → -0.632036, b → -1.36796, x0 → 1.90049},
          {a → -3.14628, b → 1.14628, x0 → -0.850887},
          {a → -1.81761, b → -0.182392, x0 → -0.130918}}

In[38]:= (* Vemos que temos soluções imaginárias,
         que não correspondem a soluções válidas para o problema em
          questão. Vamos assim restringir o NSolve a dar apenas soluções reais:  *)
```

In[39]:=
```
solucoes = NSolve[{eq1, eq2, eq3}, {a, b, x0}, Reals]
```

Out[39]=
```
{{a → 1.50863, b → -3.50863, x0 → -1.91048},
 {a → -0.632036, b → -1.36796, x0 → 1.90049},
 {a → -3.14628, b → 1.14628, x0 → -0.850887},
 {a → -1.81761, b → -0.182392, x0 → -0.130918}}
```

In[40]:=
```
(* Podemos tabelar essas soluções usando inteligentemente o comando ReplaceAll,
que abaixo é usado na forma "/.":    *)
```

In[41]:=
```
tabela = {x0, f[x0]} /. solucoes
```

Out[41]=
```
{{-1.91048, -6.39084}, {1.90049, -2.56914},
 {-0.850887, 3.82341}, {-0.130918, 0.0555663}}
```

In[42]:=
```
g3 = ListPlot[tabela, PlotStyle → {Red, Large}]
 (* gráfico com os pontos achados para o carro *)
```

Out[42]=



In[43]:=
```
Show[g1, g2, g3]
```

Out[43]=



In[44]:=
```
(* Podemos também tabelar as retas que dão
  as soluções usando novamente o comando /. abaixo   *)
```

In[45]:=
```
retas = (a x + b) /. solucoes
```
Out[45]=
```
{-3.50863 + 1.50863 x, -1.36796 - 0.632036 x, 1.14628 - 3.14628 x, -0.182392 - 1.81761 x}
```

In[46]:=
```
g4 = Plot[retas, {x, -2.5, 2.5}, PlotStyle → Orange]
 (*  gráficos com as retas encontradas  *)
```
Out[46]=


In[47]:=
```
Show[g4, g1, g2, g3]  (* gráfico mostrando tudo  *)
```
Out[47]=


In[48]:=

Bônus: uma animação com o que acontece ao longo do trajeto do carro (aperte o play):

In[49]:=
```
fig = 
```

```
;

coelho = Show[fig, ImageSize → 10];
g5 = ListPlot[{{1, -2}}, PlotMarkers → {coelho}];
```

In[52]:=
```
Animate[
 Show[g1, g5, ListPlot[{{x0, f[x0]}}, PlotStyle → {Red, Large}, PlotRange → {-7, 7}],
  Plot[f[x0] + f'[x0] (x - x0), {x, -2.5, 2.5}, PlotStyle → Orange,
   PlotRange → {-7, 7}]], {x0, -2.5, 2.5}, AnimationRunning → False]
```

Out[52]=



In[53]:=
```
(*  O mesmo que o gráfico anterior, mas com você controlando o ponto:  *)
```

In[54]:=
```
Manipulate[
 Show[g1, g5, ListPlot[{{x0, f[x0]}}, PlotStyle → {Red, Large}, PlotRange → {-7, 7}],
  Plot[f[x0] + f'[x0] (x - x0), {x, -2.5, 2.5},
   PlotStyle → Orange, PlotRange → {-7, 7}]], {{x0, 0}, -2.5, 2.5}]
```

Out[54]=